
Dotenvy is an Elixir port of the original dotenv Ruby gem, compatible with mix and releases. It is designed to help the development of applications following the principles of the 12-factor app and its recommendation to store configuration in the environment.
Installation
Add dotenvy to your list of dependencies in mix.exs:
def deps do
  [
    {:dotenvy, "~> 1.0.0"}
  ]
end
It has no dependencies.
Usage
Dotenvy is designed to help configure your application at runtime, and one
of the most effective places to do that is inside config/runtime.exs (available
since Elixir v1.11).
The Dotenvy.source/2 function can accept a single file or a list of files.  When combined with Config.config_env/0 it is easy to load up environment-specific config. A common setup in your config/runtime.exs would include a block like the following:
# config/runtime.exs
import Config
import Dotenvy

env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
  Path.absname(".env", env_dir_prefix),
  Path.absname(".#{config_env()}.env", env_dir_prefix),
  Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
  System.get_env()
])
The above example would include the envs/.env file for shared/default configuration values and then leverage environment-specific .env files (e.g. envs/.dev.env) to provide environment-specific values. An envs/.{MIX_ENV}.overrides.env file would be referenced in the .gitignore file to allow developers to override any values a file that is not under version control. System environment variables are given final say over the values via System.get_env().  Think of Dotenvy.source/2 as a merge operation, similar to Map.merge/2: the last input takes precedence.
By default, the listed files do not need to exist (you can leverage the :require_files option if needed). The Dotenvy.source/2 function only needs to know where to look. This makes it easy to commit default values while still leaving the door open to developers to override values via their own configuration files.
You control if and how existing system env vars are handled: usually they should take precedence over values defined in .env files, so for most apps, the System.get_env() should be included as the final input supplied to source/2.
Unlike other packages, Dotenvy has no opinions about the names or locations of your .env files, you just need to pass their paths to Dotenvy.source/2 or Dotenvy.source!/2.
For a simple example, we can load up a .env file containing all defaults and an environment-specific file (e.g. .dev.env).  Remember to use both Path.expand/1 and Path.absname/2 so that the operating system can resolve your file names into fully qualified paths that work when you are developing locally or when your app is running as a release or inside a Livebook.
# config/runtime.exs
import Config
import Dotenvy

env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
  Path.absname(".env", env_dir_prefix), 
  Path.absname("#{config_env()}.env", env_dir_prefix), 
  System.get_env()
  ])

config :myapp, MyApp.Repo,
    database: env!("DATABASE", :string!),
    username: env!("USERNAME", :string),
    password: env!("PASSWORD", :string),
    hostname: env!("HOSTNAME", :string!),
    pool_size: env!("POOL_SIZE", :integer),
    adapter: env!("ADAPTER", :module, Ecto.Adapters.Postgres),
    pool: env!("POOL", :module?)
And then define your variables in the file(s) to be sourced. Dotenvy has no opinions about what you name your files; .env is merely a convention.
# .env
DATABASE=myapp_dev
USERNAME=myuser
PASSWORD=mypassword
HOSTNAME=localhost
POOL_SIZE=10
POOL=
When you set up your application configuration in this way, you are creating a contract with the environment: Dotenvy.env!/2 will raise if the required variables have not been set or if the values cannot be properly transformed. This is an approach that works equally well for your day-to-day development and for mix releases.
Read the Getting Started guide for more details.
Refer to the "dotenv" (.env) file format for more examples and features of the supported syntax.
See the Dotenvy module documentation on its functions.
Note for Mix Tasks
If you have authored your own Mix tasks, you must ensure that they load the
application configuration in a way that is compatible with the runtime config.
A good way to do this is to include Mix.Task.run("app.config") inside the
run/1 implementation, e.g.
def run(_args) do
  Mix.Task.run("app.config")
  # ...
end
If you are dealing with third-party mix tasks that fail to properly load configuration, you may need to manually call mix app.config before running them, e.g.
mix do app.config other.task

Defining a task alias in mix.exs is another way to accomplish this:
# mix.exs
defp aliases do
    [
      "other.task": ["app.config", "other.task"]
    ]
Upgrading from v0.5.0 or before
Starting with Dotenvy v0.6.0, the precedence of system env variables over parsed .env files is not defined; the :overwrite? and :vars options are no longer supported in Dotenvy.source/2 and Dotenvy.source!/2. Instead, the source functions now accept file paths OR maps: this makes the question of variable precedence something that must be explicitly listed. The source functions act more like Map.merge/2, accumulating values, always giving precedence to the righthand source.
Most users upgrading from v0.5.0 will wish to include System.get_env() as the final input to source/2.
# in dotenvy 0.5.0 or before:
source(["#{config_env()}.env", "#{config_env()}.override.env"])

# should be changed to the following in dotenvy 0.6.0:
source(["#{config_env()}.env", "#{config_env()}.override.env", System.get_env()])
If you are relying on variable interpolation in your .env files, you may also need to include System.get_env() (or an equivalent subset) before you list your .env files.  This is necessary to make values available to the file parser.
# in dotenvy 0.5.0 or before:
source(["#{config_env()}.env", "#{config_env()}.override.env"])

# should be changed to the following in dotenvy 0.6.0:
source([System.get_env(), "#{config_env()}.env", "#{config_env()}.override.env", System.get_env()])
The change in syntax introduced in v0.6.0 favors a declarative list of sources over opaquely inferred inputs. This also opens the door for compatibility with other value sources, e.g. secure parameter stores.

    Cheatsheet

Quick reference for using Dotenvy.
Setup
envs/ Directory
The recommended location for storing your .env files is inside a dedicated envs/ directory.
Code editors recognize the .env extension.
Example envs/dev.env
API_KEY=xyz
AWS_REGION="us-east-1"
HTTP_CLIENT=HTTPoison
Version Control
envs/dev.env is tracked by Git. Use it to store sensible non-sensitive defaults.
envs/dev.overrides.env is ignored by Git. Use it to override any variables in the tracked version.
This must match the values used in the runtime configuration.
.gitignore
*.overrides.env
Release Compatibility
For compatibility with releases, 
configure your builds to copy (i.e. "overlay") the contents of your envs/ 
directory into the root of the release.
mix.exs
defp releases do
[
    my_app: [
      overlays: ["envs/"]
    ]
]
end
Runtime Config
Your config/runtime.exs is where you source your environment variables. 
The last argument to Dotenvy.source/2 or Dotenvy.source!/2 takes precedence.
It's common to use System.get_env() as the final argument so any existing system
environment variables will take precedence over anything parsed from the .env files.
config/runtime.exs
import Config
import Dotenvy

env_dir = System.get_env("RELEASE_ROOT") || Path.expand("./envs/")
  
source!([
    Path.absname(".env", env_dir),
    Path.absname(".#{config_env()}.env", env_dir),
    Path.absname(".#{config_env()}.overrides.env", env_dir),
    System.get_env()
  ])
Example Database Configuration
Dev ENV
envs/dev.env
PG_USERNAME=postgres
PG_PASSWORD=postgres
PG_HOSTNAME=localhost
PG_PORT=5432
PG_DATABASE=m_app_dev
PG_POOL_SIZE=10
PG_POOL=DBConnection.ConnectionPool
PG_SSL=true
Test ENV
envs/test.env
PG_USERNAME=postgres
PG_PASSWORD=postgres
PG_HOSTNAME=localhost
PG_PORT=5432
PG_DATABASE=my_app_test
PG_POOL_SIZE=10
PG_POOL=Ecto.Adapters.SQL.Sandbox
PG_SSL=false
Runtime Configuration
config/runtime.exs
import Config
import Dotenvy

env_dir = System.get_env("RELEASE_ROOT") || Path.expand("./envs")
  
source!([
    Path.absname(".env", dir),
    Path.absname(".#{config_env()}.env", env_dir),
    Path.absname(".#{config_env()}.overrides.env", env_dir),
    System.get_env()
  ])

config :my_app, MyApp.PGRepo,
  pool: env!("PG_POOL", :module?),
  pool_size: env!("PG_POOL_SIZE", :integer),
  database: env!("PG_DATABASE", :string),
  username: env!("PG_USERNAME", :string),
  password: env!("PG_PASSWORD", :string),
  port: env!("PG_PORT", :integer),
  hostname: env!("PG_HOSTNAME", :string)
Transformations
System Environment variables are always stored as strings which may need to be 
transformed into native Elixir data types.
Used as the 2nd argument to Dotenvy.env!/2 and Dotenvy.env!/3
	Conversion Type	Elixir Type	On Empty String
	:atom	atom	:""
	:atom?	atom	nil
	:atom!	atom	raise ⚠
	:boolean	boolean	false
	:boolean?	boolean	nil
	:boolean!	boolean	raise ⚠
	:charlist	charlist	'' i.e. []
	:charlist?	charlist	nil
	:charlist!	charlist	raise ⚠
	:integer	integer	0
	:integer?	integer	nil
	:integer!	integer	raise ⚠
	:float	float	0
	:float?	float	nil
	:float!	float	raise ⚠
	:existing_atom	atom	:"" or raise
	:existing_atom?	atom	nil
	:existing_atom!	atom	raise ⚠
	:module	atom	:"Elixir."
	:module?	atom	nil
	:module!	atom	raise ⚠
	:string	String	""
	:string?	String	nil
	:string!	String	raise ⚠

Custom functions handle their own behavior.
Custom Function Example
PHX_IP="0, 0, 0, 0, 0, 0, 0, 0"
config :feenix, FeenixWeb.Endpoint,
    http: [
      # Enable IPv6 and bind on all interfaces.
      ip: env!("PHX_IP", fn ip -> 
        ip 
        |> String.split(",") 
        |> Enum.map(&String.trim/1) 
        |> Enum.map(&String.to_integer/1) 
        |> List.to_tuple()
      end)
    ],
Your custom functions can raise a Dotenvy.Error to benefit from improved messages
that include helpful context about any problems, e.g.
strict_boolean! = fn
  "true" -> true
  "false" -> false
  _ ->
    raise Dotenvy.Error,
      message: "strict_boolean! values must be either true or false"
end

config :myapp, :some_bool, env!("SOME_BOOL", strict_boolean!)
This will yield an error like the following:
** (RuntimeError) Error converting variable SOME_BOOL using custom function: strict_boolean! values must be either true or false


  

    Changelog

The format is based on Keep a Changelog,
and this project adheres to Semantic Versioning.
v1.1.1
	Fixes compiler warnings in Elixir 1.19
	Bumps deps to latest

v1.1.0
	Adds support for Dotenvy.Parser.parse/3 options :sys_cmd_fn and :sys_cmd_opts to
allow for greater control over how (or if) any system commands (designated via $())
are executed during parsing.
	Updates dependencies to latest
	Docs cleanup around RELEASE_ROOT so the examples tolerate cases where the path
includes or omits a trailing slash. This is important because a bad path there causes
all kinds of frustration in a release.

v1.0.1
	Minor docs cleanup re Markdown syntax.
	Added link for Sponsor

v1.0.0
	Adds support for shell commands via the $() syntax
	new documentation pages added to introduce generators and more examples
	Deletes deprecated Dotenvy.env/3 function
	Updates dependencies to latest
	Better error messages for unset variables
	Better error messages for :require_files errors
	Updated Logo

v0.9.0
	Removes fallback to System.fetch_env/1 and instead requires explicit sourcing of system envs. See Issue 21
	Bumps Elixir version specified for local development in .tool-versions
	Updates dependencies to latest

v0.8.0
	Enables exception rescuing to report on problems with custom callback functions
	Moves Dotenvy.Transformer.Error to Dotenvy.Error to offer a simpler interface
for devs who want to raise errors from custom transformer functions
	Improved documentation and examples for usage in umbrella apps
	Updates all internal options in the parser to use %Opts{} struct
	Improves test coverage
	Bumps Elixir version specified for local development in .tool-versions
	Updates dependencies to latest

v0.7.0
	Formally defines a type for all supported conversions to improve documentation and specs
	Updates dependencies to latest
	Specifies Elixir 1.13 as required (simply because I can't get anything older to compile)

v0.6.0
	Does away with the confusing :overwrite? and vars options in favor of a simple declarative/explicit inputs. source/2 now accepts ad-hoc maps as inputs.
	Updates dependencies including :ex_doc to take advantage of admonishment blocks.
	Various documentation cleanups/clarifications.

v0.5.0
	Shifts storage of system environment variables to the application process dictionary and alters the reading of this data to help improve the security posture and avoid leaking env values. :side_effect option for source/2 and source!/2 function changed.

v0.4.1
	Makes error messages more informative when unable to convert strings to integers or floats

v0.4.0
	Adds support for custom transformer types by allowing an arity 1 function as the second argument to Dotenvy.Transformer.to/2. See Issue 2

v0.3.0
	Renames Dotenvy.Transformer.to/2 to Dotenvy.Transformer.to!/2 to better communicate that it may raise an error.
	Returns key name in errors for easier troubleshooting.
	Tracks an error if the :require_files option lists a file not included in the files input (for sanity).
	Introduces Dotenvy.env!/3 (which is the same as Dotenvy.env/3 but with no defaults provided). This better communicates that it may raise an error (because internally it relies on Dotenvy.Transformer.to!/2)
	Deprecates Dotenvy.env/3 in favor of Dotenvy.env!/3

v0.2.0
Adds support for default type of :string to the Dotenvy.env!/2 and Dotenvy.env/3 functions.
v0.1.0
Initial release.


  

    Getting Started

The concept of environment variables is simple and Dotenvy aims to make your application take advantage of them, but how can you start using them easily in your application?  This page will walk you through kicking the tires of a simple application so you can learn how Dotenvy works.
Prerequisite
If you haven't already, install the dotenvy_generators.
When you run mix help, you should see dot.new as one of the available tasks. Make sure that's available before continuing.

Generating an app
The dot.new mix task is available when you have installed the dotenvy_generators. We can use it generate a new Elixir app:
mix dot.new example
Follow the prompts given:
cd example
mix deps.get
The structure should look familiar to you -- the only thing you might notice is the presence of the envs/ directory.
Environment-specific env files
Take a look at the config/runtime.exs. It includes a line like the following which reads from an environment variable named SECRET:
config :example, :secret, env!("SECRET", :string!)
If you start your app using iex -S mix and enter into the iex shell:
iex> Application.get_env(:example, :secret)
"my-secret-dev"
This value came from the envs/.dev.env file, which declares the following:
SECRET=my-secret-dev
Next, let's try running a test. Open up the test/example_test.exs file and edit it so we have single test like the following:
test "reads variables specific to an env" do
    assert "my-secret-dev" == Application.get_env(:example, :secret)
end
Then run mix test.  
Uh oh! The test fails because Application.get_env(:example, :secret) returned "my-secret-test" and not "my-secret-dev".
Take a look at the envs/.test.env file and see how the variable is declared there. We can now adjust our test so the assertion matches the value we declared in our .test.env:
test "reads variables specific to an env" do
    assert "my-secret-test" == Application.get_env(:example, :secret)
end
Running mix test now passes!
Core Concept: variables are read from an environment-specific file
Just like with Elixir's regular config files, Dotenvy loads the appropriate
env file depending on your environment. Look at how the config_env() function
is used in runtime.exs to determine the file name; different values are
declared in the .test.env and .dev.env.
Environment Variables
Next, let's try to access the environment variable directly:
iex> System.get_env("SECRET")
nil
What happened? Application.get_env(:example, :secret) worked, so why doesn't System.get_env("SECRET") see the variable?
The answer to this riddle is that Dotenvy is read-only: Dotenvy does not set environment variables. This helps keep things locked down. It may be counter-intuitive, but Dotenvy doesn't even necessarily read environment variables!  Dotenvy only reads the inputs you give it. Dotenvy only reads environment variables if you pass it the output from System.get_env().
Core Concept: Dotenvy does not set ENV vars
Any variables you declare in your from in your env files are not exported
back to the system; i.e. System.put_env/2 is NOT called. In other words,
declaring a variable FOO in one of your parsed .env files does not mean
System.get_env/2 can be used to retrieve it later. This encapsulation is by design!
If you want to set environment variables, you must do it explicitly.
Establishing a contract
One of the tenets of the 12-factor App is to have a clean contract with the underlying operating system, offering maximum portability between execution environments. Our runtime.exs is largely responsible for this: it dictates exactly which variables it needs.
To see this in action, let's add another configuration setting to our app by adding the following line to our example runtime.exs:
config :example, :password, env!("PASSWORD", :string!)
Then stop and restart your app by pressing ctrl-C and running iex -S mix once more.  You will see an error:
** (RuntimeError) Environment variable PASSWORD not set
The application is declaring its contract by specifying that certain environment variables must be present. Because the PASSWORD variable is not set, the application will not start because the contract has not been met.
We can provide the variable on the command line:
    PASSWORD=xxxx iex -S mix
And the app will start normally.
Alternatively, you can provide this value in your env files. Add the following to your envs/.dev.exs file:
    PASSWORD=xxxx
Your application will now start in the dev environment. However if you try to run tests, you will once again see the RuntimeError. You can rectify this by supplying a value in the envs/.test.dev file.
A good convention here is to have a default .env file loaded first which lists all the variables that your app needs. That's a great place to put some documentation too!
Core Concept: your app should dictate which variables it needs
if your application needs certain configuration values to run, then it should
demand that those values are set.  The implication is that if those values aren't
there, there's no point in starting the app because it can't do what it needs to do.
This is the contract with the environment.
Type-casting
All environment variables store string values. Dotenvy.env!/2 and Dotenvy.env!/3 have as their second argument an atom which determines how to convert the string value. For example, you may need to convert a PORT variable into an integer, other values may need to be booleans, and others may need to be atoms or modules.
There is some subtlety involved here when it comes to how empty values should be handled.
Let's revisit our PASSWORD variable from the previous section. Let's try setting the environment variable to an empty value before we start our app:
$ PASSWORD= iex -S mix
** (RuntimeError) Error converting variable PASSWORD to string!: non-empty value required

Uh-oh! This causes another error, this time not because the variable wasn't set (it was), but because its value was empty.
Let's modify the line in our runtime.exs and replace :string! with :string (i.e. remove the exclamation point):
config :example, :password, env!("PASSWORD", :string)
Now the application starts fine, even if the PASSWORD value is empty. Probably a password needs to have a value, so using :string! for the second argument is probably more appropriate, but you can decide this on a case-by-case basis.
Understanding type-casting is another core concept in helping to leverage Dotenvy so  your app get what it needs to run.
Core Concept: type-casting
For each variable you read via Dotenvy.env!/2 in config/runtime.exs, you
should consider what the resulting Elixir value needs to be. Can the value be empty? Are
nil values allowed? Choose the conversion type
that best supplies your app with the value it needs.
See the section on releases for further information on how Dotenvy works in the context of a Mix release.


  

    Releases

One of the hurdles when dealing with Elixir releases is that only certain files are packaged into them.  Any new ad-hoc files like our .env files are not included by default.  One way to ensure that our additional files get packaged into the release is to specify the :overlays option in your mix.exs. To do this we edit the mix.exs file to specify the envs/ directory which contains your .env files:
# mix.exs

def project do
  [
    app: :your_app,
    # ... 
    releases: releases()
  ]
end

defp releases do
  [
    your_app: [
        overlays: ["envs/"]
    ]
  ]
end
Core Concept: Overlays
When you specify a folder in the overlays option in your mix.exs, then the
contents (and not the folder itself) will be copied to the root of the release.

During day-to-day development, your .env files live inside the envs/ folder, but when a release is built, they get copied to the root of the release, so we cannot rely on relative paths in runtime.exs!  In our examples, we rely on Path.expand/2, Path.absname/2, and the presence of the RELEASE_ROOT system environment variable to resolve our relative paths into absolute paths.
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")
This is a simple trick to ensure that we always have a fully-qualified path to where our .env files live. This pattern is repeated throughout the documentation because it is so important!
Putting your .env files inside a folder named envs/ is merely a convention: you are free to store them where you wish, but keep in mind that it is easier to deal with folders than it is with individual files.  See the documentation on Mix Release Overlays for more information.
Our config/runtime.exs will look something like the following. Note that the folder referenced in the mix.exs overlays section (envs/) must correspond with the path referenced in config/runtime.exs.
import Config
import Dotenvy

# For local development, read dotenv files inside the envs/ dir;
# for releases, read them at the RELEASE_ROOT
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ])
Remember that is always safer to use an absolute path. This is especially important when working with umbrella apps or Livebooks!
Umbrella Apps
Elixir Umbrella Projects consume configuration slightly differently due to how they are organized.
In particular, you have to be very careful about relative paths when working in an umbrella project. Depending on what you're doing, the path may be relative to a single application instead of relative to the root of the repository. As elsewhere, using Path.expand/1 and Path.absname/2 is a good way to anchor your config/runtime.exs so it resolves to the same directory no matter if the app is running locally or as a release.
Once again, the winning pattern for your config/runtime.exs will look something like this:
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ])
Changing the envs/ folder
What if you wish to keep your .env files in some other folder?  No problem. You just need to update your runtime.exs and your mix.exs so the :overlays option corresponds to the folder name.
For example, here's what you would do if wanted to keep your .env files inside a directory named xyz:
# config/runtime.exs
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./xyz")
# ... etc...
# mix.exs
  defp releases do
    [
      my_app: [
        overlays: ["xyz/"]
      ]
    ]
  end
Some languages/frameworks store .env files at the root of the application, but this isn't easily compatible with Elixir releases.  Rather than trying to push the river, we recommend choosing a sub-folder and leveraging the :overlays option.


  

    Phoenix

This page shows you how to either generate a new Phoenix application using the dotenvy_generators package OR how to retrofit the config files of an existing Phoenix app.
Creating a new Phoenix application that uses Dotenvy
Make sure you have installed the dotenvy_generators before continuing!
In a new terminal window, you can run the new task to generate a new Phoenix app, e.g. mix phx.new.dotenvy hello.  This should generate a functional Phoenix application that leverages Dotenvy for its configuration.
Have a look over the file structure: notice the envs/ directory.  The files there house the values read at runtime, whereas the various config files inside of config/ have been cleaned up so they focus on providing settings that must be defined at compile-time.
Manually Editing Files
If you have an existing Phoenix application and you want to modify it to use Dotenvy, then you can reference the files below as a guideline for editing your configuration files.
Pay attention to how the files are organized: most of the configuration has been moved into the runtime.exs leaving only minimal bits in the env-specific compile-time configs. Remember that one of the guiding principles of Dotenvy is to use runtime configuration whenever possible.
Replace the values to match your app
You will need to replace YourApp, YourAppWeb, and :your_app with
the appropriate modules and app name for your application. Don't just copy
and paste these sample files!  And make sure you don't overwrite any existing config
for other apps/services that might not be present in this example.
config/config.exs
# config/config.exs
# This file is responsible for configuring your application
# and its dependencies with the aid of the Config module.
#
# This configuration file is loaded before any dependency and
# is restricted to this project.

# General application configuration
import Config

config :your_app,
  ecto_repos: [YourApp.Repo],
  generators: [timestamp_type: :utc_datetime]

# Configures the endpoint
config :your_app, YourAppWeb.Endpoint,
  adapter: Bandit.PhoenixAdapter,
  render_errors: [
    formats: [html: YourAppWeb.ErrorHTML, json: YourAppWeb.ErrorJSON],
    layout: false
  ],
  pubsub_server: YourApp.PubSub,
  live_view: [signing_salt: "xyz123AB"]

# Configure esbuild (the version is required)
config :esbuild,
  version: "0.17.11",
  your_app: [
    args:
      ~w(js/app.js --bundle --target=es2017 --outdir=../priv/static/assets --external:/fonts/* --external:/images/*),
    cd: Path.expand("../assets", __DIR__),
    env: %{"NODE_PATH" => Path.expand("../deps", __DIR__)}
  ]

# Configure tailwind (the version is required)
config :tailwind,
  version: "3.4.3",
  your_app: [
    args: ~w(
      --config=tailwind.config.js
      --input=css/app.css
      --output=../priv/static/assets/app.css
    ),
    cd: Path.expand("../assets", __DIR__)
  ]

# Configures Elixir's Logger (compile-time config)
config :logger, :console,
  format: "$time $metadata[$level] $message\n",
  metadata: [:request_id]

# Import environment specific config. This must remain at the bottom
# of this file so it overrides the configuration defined above.
import_config "#{config_env()}.exs"
config/dev.exs
# config/dev.exs
import Config

# Compile-time configuration includes code_reloader, debug_errors, and force_ssl
# https://hexdocs.pm/phoenix/Phoenix.Endpoint.html#module-compile-time-configuration
config :your_app, YourAppWeb.Endpoint,
  code_reloader: true,
  debug_errors: true,
  force_ssl: false,
  # Watch static and templates for browser reloading.
  live_reload: [
    patterns: [
      ~r"priv/static/(?!uploads/).*(js|css|png|jpeg|jpg|gif|svg)$",
      ~r"priv/gettext/.*(po)$",
      ~r"lib/your_app_web/(controllers|live|components)/.*(ex|heex)$"
    ]
  ],
  # The watchers configuration can be used to run external
  # watchers to your application. For example, we can use it
  # to bundle .js and .css sources.
  # Watchers can be configured at runtime but are unlikely to change
  watchers: [
    esbuild: {Esbuild, :install_and_run, [:your_app, ~w(--sourcemap=inline --watch)]},
    tailwind: {Tailwind, :install_and_run, [:your_app, ~w(--watch)]}
  ]

# Compile-time config
config :phoenix_live_view,
  debug_heex_annotations: true,
  enable_expensive_runtime_checks: true

# Enable dev routes for dashboard and mailbox (compile-time config)
config :your_app, dev_routes: true

# Do not include metadata nor timestamps in development logs
config :logger, :console, format: "[$level] $message\n"
config/test.exs
# config/test.exs
import Config

# Print only warnings and errors during test
config :logger, level: :warning
config/prod.exs
import Config

# Compile-time configuration includes code_reloader, debug_errors, and force_ssl
# https://hexdocs.pm/phoenix/Phoenix.Endpoint.html#module-compile-time-configuration
config :your_app, YourAppWeb.Endpoint,
  code_reloader: false,
  debug_errors: false,
  force_ssl: false

# Do not print debug messages in production
config :logger, level: :info
config/runtime.exs
# config/runtime.exs
import Config
import Dotenvy

# For local development, read dotenv files inside the envs/ dir;
# for releases, read them at the RELEASE_ROOT
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!(
  [
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ],
  require_files: [Path.absname(".env", env_dir_prefix)]
)


# config/runtime.exs is executed for all environments, including
# during releases. It is executed after compilation and before the
# system starts. Do not define any compile-time configuration in here,
# as it won't be applied.

# ## Using releases
#
# If you use `mix release`, you need to explicitly enable the server
# by passing the PHX_SERVER=true when you start it:
#
#     PHX_SERVER=true bin/your_app start
#
# Alternatively, you can use `mix phx.gen.release` to generate a `bin/server`
# script that automatically sets the env var above.
if env!("PHX_SERVER", :boolean!) do
  config :your_app, YourAppWeb.Endpoint, server: true
end

# Initialize plugs at runtime for faster development compilation
# values can be :runtime or :compile; must be :compile in prod (the default)
config :phoenix, :plug_init_mode, env!("PHX_PLUGIN_INIT_MODE", :existing_atom!)

# Use Jason for JSON parsing in Phoenix
config :phoenix, :json_library, Jason

ip =
  env!("HTTP_INTERFACE", fn val ->
    with [_] <- String.split(val, "."),
         [_] <- String.split(val, ":") do
      raise "Invalid IP address specified"
    else
      parts -> parts |> Enum.map(&String.to_integer/1) |> List.to_tuple()
    end
  end)

ecto_socket_options = if env!("ECTO_IPV6", :boolean!), do: [:inet6], else: []

config :your_app, YourApp.Repo,
  # ssl: true,
  url: env!("DATABASE_URL", :string!),
  pool: env!("PG_POOL", :module?),
  pool_size: env!("POOL_SIZE", :integer!),
  socket_options: ecto_socket_options,
  stacktrace: env!("ECTO_STACKTRACE", :boolean),
  show_sensitive_data_on_connection_error:
    env!("SHOW_SENSITIVE_DATA_ON_CONNECTION_ERROR", :boolean)

if env!("ENABLE_DISTRIBUTED_MODE", :boolean) do
  config :your_app, :dns_cluster_query, env!("DNS_CLUSTER_QUERY", :string)
end

config :your_app, YourAppWeb.Endpoint,
  cache_static_manifest: env!("PHX_CACHE_STATIC_MANIFEST", :string?),
  check_origin: env!("HTTP_CHECK_ORIGIN", :boolean),
  http: [ip: ip, port: env!("PORT", :integer!)],
  secret_key_base: env!("SECRET_KEY_BASE", :string!),
  # Used to build URLs
  url: [
    host: env!("PHX_HOST", :string!),
    port: env!("PHX_URL_PORT", :integer!),
    scheme: env!("PHX_URL_SCHEME", :string!)
  ]

# ## SSL Support
#
# To get SSL working, you will need to add the `https` key
# to your endpoint configuration:
#
#     config :your_app, YourAppWeb.Endpoint,
#       https: [
#         ...,
#         port: 443,
#         cipher_suite: :strong,
#         keyfile: System.get_env("SOME_APP_SSL_KEY_PATH"),
#         certfile: System.get_env("SOME_APP_SSL_CERT_PATH")
#       ]
#
# The `cipher_suite` is set to `:strong` to support only the
# latest and more secure SSL ciphers. This means old browsers
# and clients may not be supported. You can set it to
# `:compatible` for wider support.
#
# `:keyfile` and `:certfile` expect an absolute path to the key
# and cert in disk or a relative path inside priv, for example
# "priv/ssl/server.key". For all supported SSL configuration
# options, see https://hexdocs.pm/plug/Plug.SSL.html#configure/1
#
# We also recommend setting `force_ssl` in your config/prod.exs,
# ensuring no data is ever sent via http, always redirecting to https:
#
#     config :your_app, YourAppWeb.Endpoint,
#       force_ssl: [hsts: true]
#
# Check `Plug.SSL` for all available options in `force_ssl`.
#
# In order to use HTTPS in development, a self-signed
# certificate can be generated by running the following
# Mix task:
#
#     mix phx.gen.cert
#
# Run `mix help phx.gen.cert` for more information.
#
# The `http:` config above can be replaced with:
#
#     https: [
#       port: 4001,
#       cipher_suite: :strong,
#       keyfile: "priv/cert/selfsigned_key.pem",
#       certfile: "priv/cert/selfsigned.pem"
#     ],
#
# If desired, both `http:` and `https:` keys can be
# configured to run both http and https servers on
# different ports.

# ## Configuring the mailer
#
# In production you need to configure the mailer to use a different adapter.
# Also, you may need to configure the Swoosh API client of your choice if you
# are not using SMTP. Here is an example of the configuration:
#
#     config :your_app, YourApp.Mailer,
#       adapter: Swoosh.Adapters.Mailgun,
#       api_key: System.get_env("MAILGUN_API_KEY"),
#       domain: System.get_env("MAILGUN_DOMAIN")
#
# For this example you need include a HTTP client required by Swoosh API client.
# Swoosh supports Hackney and Finch out of the box:
#
#     config :swoosh, :api_client, Swoosh.ApiClient.Hackney
#
# See https://hexdocs.pm/swoosh/Swoosh.html#module-installation for details.
# Configures Swoosh API Client
if env!("SWOOSH_API_CLIENT", :boolean) do
  config :swoosh, api_client: env!("SWOOSH_API_CLIENT", :module!), finch_name: YourApp.Finch
else
  config :swoosh, api_client: false
end

config :your_app, YourApp.Mailer, adapter: env!("SWOOSH_MAILER_ADAPTER", :module)
config :swoosh, api_client: Swoosh.ApiClient.Finch, finch_name: YourApp.Finch
config :swoosh, local: env!("SWOOSH_LOCAL_MEMORY_STORAGE", :boolean)
envs/.env
The shared/default values can be listed and documented here.  The .dev.env and other .env files can either copy this file in its entirety and modify the values (for easier diff comparisons), or they can include only the variables that they need to change.
# envs/.env (default/shared config)
# ###################
# Distributed Node  #
# ###################
# If distributed mode is enabled, the DNS_CLUSTER_QUERY variable must have a value
ENABLE_DISTRIBUTED_MODE=false
DNS_CLUSTER_QUERY=

# #####################
# Phoenix / Webserver #
# #####################
# The interface(s) to listen on.
# Can be specified as the following formats:
# `1.2.3.4` for IPv4 addresses (using period separators)
# `1:2:3:4:5:6:7:8` for IPv6 addresses (using colon separators)
# 
# To bind to loopback IPv4 address & prevent access from other machines: `127.0.0.1`
# To allow access from other machines: `0.0.0.0`
# 
# To enable IPv6 and bind on all interfaces: `0:0:0:0:0:0:0:0`
# For local network only access: `0:0:0:0:0:0:0:1`
# 
# See the documentation on https://hexdocs.pm/bandit/Bandit.html#t:options/0
# for details about using IPv6 vs IPv4 and loopback vs public addresses.
HTTP_INTERFACE=127.0.0.1
HTTP_CHECK_ORIGIN=false
# Port where HTTP requests will be accepted
PORT=4000
# A secret key used as a base to generate secrets for encrypting and signing data. 
# For example, cookies and tokens are signed by default, but they may also be 
# encrypted if desired. Must be set per application.
SECRET_KEY_BASE=
# Boolean indicating whether to start the Phoenix webserver implicitly.
# Usually this is false for dev and true for prod
PHX_SERVER=false
# PHX_HOST, PHX_URL_PORT, and PHX_URL_SCHEME are used to create links
PHX_HOST=localhost
# Duplicate the ${PORT} value for http; 443 recommended for https
PHX_URL_PORT=
# scheme may be either http or https
PHX_URL_SCHEME=http
# Specifies the path to a cache manifes containing the digested version of 
# static files. This manifest is generated by the `mix assets.deploy` task,
# which you should run after static files are built and
# before starting your production server.
# Leave empty for dev or when you do not need a manifest
PHX_CACHE_STATIC_MANIFEST=
# Set a higher stacktrace during development, e.g. 20, or set to false to disable.
# Disabling is recommended in prod as building large stacktraces may be expensive.
PHX_STACKTRACE_DEPTH=20
# Initialize plugs at runtime for faster development compilation
# values can be :runtime or :compile; must be :compile in prod (the default)
PHX_PLUGIN_INIT_MODE=compile

############
# Database #
############
# The MIX_TEST_PARTITION environment variable can be used
# to provide built-in test partitioning in CI environment.
# Run `mix help test` for more information.
MIX_TEST_PARTITION=
PG_USERNAME=postgres
PG_PASSWORD=postgres
PG_HOSTNAME=localhost
PG_DATABASE=your_app_dev
PG_POOL=DBConnection.ConnectionPool
POOL_SIZE=10
# DATABASE_URL format is `ecto://USER:PASS@HOST/DATABASE`
DATABASE_URL=ecto://${PG_USERNAME}:${PG_PASSWORD}@${PG_HOSTNAME}/${PG_DATABASE}
SHOW_SENSITIVE_DATA_ON_CONNECTION_ERROR=true
ECTO_IPV6=false
ECTO_STACKTRACE=true

###################
# Swoosh / Mailer #
###################
# Specify a module as a Swoosh API client for production adapters or set to false
# to disable. See https://hexdocs.pm/swoosh/Swoosh.ApiClient.html
SWOOSH_API_CLIENT=false
SWOOSH_LOCAL_MEMORY_STORAGE=true
# By default it uses the Swoosh.Adapters.Local adapter which stores the emails
# locally. You can see the emails in your browser, at "/dev/mailbox".
# For production it's recommended to configure a different adapter
SWOOSH_MAILER_ADAPTER=Swoosh.Adapters.Local
envs/.dev.env
# envs/.dev.env
HTTP_INTERFACE=127.0.0.1
HTTP_CHECK_ORIGIN=false
PORT=4000
SECRET_KEY_BASE=6D1kPrXseBg9F1O9cZUd8ocIH7l3OgT9ZlopIXqDr+jYjNrcvbjqwNvzHPKIakcF
PHX_URL_PORT=${PORT}
PHX_PLUGIN_INIT_MODE=runtime

PG_USERNAME=postgres
PG_PASSWORD=postgres
PG_HOSTNAME=localhost
PG_DATABASE=your_app_dev
PG_POOL=DBConnection.ConnectionPool
POOL_SIZE=10
# DATABASE_URL format is `ecto://USER:PASS@HOST/DATABASE`
DATABASE_URL=ecto://${PG_USERNAME}:${PG_PASSWORD}@${PG_HOSTNAME}/${PG_DATABASE}
SHOW_SENSITIVE_DATA_ON_CONNECTION_ERROR=true
ECTO_IPV6=false
ECTO_STACKTRACE=true
envs/.test.env
# envs/.test.env
HTTP_INTERFACE=127.0.0.1
HTTP_CHECK_ORIGIN=false
PORT=4002
SECRET_KEY_BASE=514niO7SLGRKi01EN1fQUqekoyfHSQL0640m64tAkJSPG7anLA6iPxWrcIUzCgyA
PHX_URL_PORT=${PORT}
PHX_PLUGIN_INIT_MODE=runtime

PG_USERNAME=postgres
PG_PASSWORD=postgres
PG_HOSTNAME=localhost
PG_DATABASE=your_app_test
PG_POOL=Ecto.Adapters.SQL.Sandbox
POOL_SIZE=10
# DATABASE_URL format is `ecto://USER:PASS@HOST/DATABASE`
DATABASE_URL=ecto://${PG_USERNAME}:${PG_PASSWORD}@${PG_HOSTNAME}/${PG_DATABASE}
SHOW_SENSITIVE_DATA_ON_CONNECTION_ERROR=true
ECTO_IPV6=false
ECTO_STACKTRACE=true

SWOOSH_MAILER_ADAPTER=Swoosh.Adapters.Test
envs/.prod.env
In prod, you may have certain env variables provided by your host.  For example. Fly.io will define a number of env variables for you. It can be helpful to list them in your .prod.env file as a reminder.
# envs/.prod.env
ENABLE_DISTRIBUTED_MODE=true
# DNS_CLUSTER_QUERY= # set by Fly.io

HTTP_INTERFACE=0:0:0:0:0:0:0:0
HTTP_CHECK_ORIGIN=true
# PORT= # set by Fly.io
# SECRET_KEY_BASE= # Set by Fly.io
PHX_SERVER=true
PHX_HOST=your-app-website.com
PHX_URL_PORT=443
PHX_URL_SCHEME=https
PHX_CACHE_STATIC_MANIFEST=priv/static/cache_manifest.json
PHX_STACKTRACE_DEPTH=false
PHX_PLUGIN_INIT_MODE=compile

# DATABASE_URL= # set by Fly.io
SHOW_SENSITIVE_DATA_ON_CONNECTION_ERROR=false
ECTO_IPV6=true
ECTO_STACKTRACE=false

SWOOSH_API_CLIENT=Swoosh.ApiClient.Finch
SWOOSH_LOCAL_MEMORY_STORAGE=false
# For production it's recommended to configure a different adapter...
# SWOOSH_MAILER_ADAPTER=Swoosh.Adapters.Local


  

    Minimal Setup

Here is another demonstration of how you can configure your app for maximum simplicity: less is more.  Per the 12-Factor App, this strategy helps "Minimize divergence between development and production". In some cases, the dev, prod, and test versions of the app can be identical (even down the the md5 hash) because the only differences are the configuration values supplied at runtime.
To see this strategy in action, we will abolish the dev.ex, prod.exs, and test.exs entirely and leave only the compile-time config (config/config.exs) and the runtime configuration (config/runtime.exs).
To try this out:
	Use the dot.new generator to generate a new application, e.g. mix dot.new sparse
	Delete the env-specific compile-time configs: config/dev.exs, config/prod.exs, and config/test.exs.
	Remove the line at the end of config/config.exs that uses import_config to load these other files.  
	If environment-specific compile-time configurations are needed, update yourconfig/config.exs to include blocks for each config environment.

Here is an example of the stripped down config/config.exs file:
config/config.exs
# compile-time config
import Config

# Dev
if config_env() == :dev do
    config :logger, :console, 
        level: :debug,
        format: "[$level] $message\n"
end

# Test
if config_env() == :test do
    config :logger, :console, 
            level: :warning,
            format: "[$level] $message\n"
end

# Prod
if config_env() == :prod do
    config :logger, :console,
        format: "$time $metadata[$level] $message\n",
        level: :info,
        metadata: [:request_id]
end
Runtime vs. Compile-time Considerations
Logger configuration is an interesting example -- because the Logger relies on
macros, it is affected by compile-time considerations. Although you can configure
the logger level at runtime, that has a different effect than configuring it at
compile-time.  If you set the logger level to :info at runtime, then you will only
see info, warning, or error messages in the logs... but the important thing is that
the calls to Logger.debug/2 are still there in the code. Calls to those functions
are still made; only the output is silenced.
By comparison, if you set the logger level to :info in the compile-time config, all
calls to Logger.debug/2 are removed from the compiled application.  Calls are no
longer made to Logger.debug/2 because that function no longer exists. That
distinction may not matter for small apps, but you can imagine that those
milliseconds can add up for mission-critical applications where performance is
paramount.
TL;DR: Sometimes it is better to have the flexibility to change the logging level at
runtime (e.g. to help debug some tricky problem) and sometimes it's better to control
this at compile-time. You decide.
Phoenix relies on macros to generate routes, so you must configure certain things at compile time: its routes are compiled into existence. Consider the following bits of configuration from a Phoenix application:
# Compile-time config
config :phoenix_live_view,
  debug_heex_annotations: true,
  enable_expensive_runtime_checks: true

# Enable dev routes for dashboard and mailbox (compile-time config)
config :your_app, dev_routes: true
Those must be defined at compile-time because it controls how the application gets built.  You can put these types of configuration details into the appropriate environment block in the compile-time config.exs.
config/runtime.exs
Your runtime.exs can include whatever it needs to, i.e. all the application configuration that can happen at runtime and you know that the calls to Dotenv.source/2 will be responsible for loading up the proper .env files for the given environment.
import Config
import Dotenvy

env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ])


config :myapp, MyApp.Repo,
  database: env!("DATABASE", :string!),
  username: env!("USERNAME", :string),
  password: env!("PASSWORD", :string),
  hostname: env!("HOSTNAME", :string!),
  pool_size: env!("POOL_SIZE", :integer),
  adapter: env("ADAPTER", :module, Ecto.Adapters.Postgres),
  pool: env!("POOL", :module?)

  # etc...
Optimizing Compilation Time
By default, your Elixir application compiles differently between environments, so you end up with multiple compiled artifacts in your _build directory.  All this extra time spent compiling and re-compiling can really add up, so it's worth asking whether or not it's necessary.
If you have an app that can be fully configured at runtime and there are no differences between the compiled versions, then you can set the :build_per_environment option in yourmix.exs so that all environments use the same compiled code and you no longer need to re-compile your app between environments.
  def project do
    [
      app: :my_app,
      version: "0.1.0",
      elixir: "~> 1.18",
      start_permanent: Mix.env() == :prod,
      deps: deps(),
      releases: releases(),
      build_per_environment: false
    ]
  end
In some cases, you may be able to leverage this even if there are differences between dev and prod, but you'll have to explore the subtleties with your particular app. For example, if dev and test can share the same compiled artifacts, that may never conflict with the prod artifacts which may only need to be compiled on a build machine as part of your deployment pipeline. YMMV.
See Mix Project configuration for the official docs.
See Also
See the article about using Dotenvy in your application for more discussion on this and other approaches.


  

    Livebooks

When running an Elixir application from a Livebook, we run into a familiar problem of resolving relative file names. The .env path that made sense when you started your app from your application root using mix will not work when starting the app from a Livebook server executing in some faraway directory.
In order to overcome this problem, it is important to resolve relative paths into absolute paths.  As demonstrated elsewhere, this can be done using Path.expand/1 and Path.absname/2:
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!([
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ])
Now your Livebooks can install apps using Dotenvy as you would other applications. In your Livebook setup, include a Mix.install/2 block like the following:
Mix.install(
  [
    {:app_using_dotenvy, path: "/path/to/app_using_dotenvy", env: :dev}
  ],
  config_path: :app_using_dotenvy,
  lockfile: :app_using_dotenvy
)


  

    Fly.io

Dotenvy can be used when deploying an app to Fly.io; other hosting services may require customizations similar to the ones demonstrated here. Following the instructions on the Phoenix page will get you most of the way there, but you also need to modify the Dockerfile so it copies over your envs/ directory: the .env files must be available to the release command. The default Fly.io Dockerfile needs a small alteration to ensure that your .env files are along for the ride.
Dockerfile
A Dockerfile gets generated when you setup your app (it is ultimately generated from mix phx.gen.release --docker). The Dockerfile is ultimately what is responsible for running mix release, so you need to ensure that all of your files and folders are copied into the container.
In the examples provided in this documentation, .env files are stored in a directory named envs/.  If that's the directory housing your .env files, then you will need to ensure that folder gets copied into the Docker container so it can properly build the release.  To do this, you will need to add a line COPY envs envs before the Run mix release command.
# ... existing Docker stuff...

# Compile the release
RUN mix compile

# Changes to config/runtime.exs don't require recompiling the code
COPY config/runtime.exs config/

COPY rel rel

# <---- make the directory of envs available to the release!!!
COPY envs envs

RUN mix release

# ... existing Docker stuff cont'd...
Once that line is there, then running fly deploy and other flyctl commands should be able to build the release and include the .env files as expected. You can test the standard GitHub workflows by pushing changes to main.
The importance of path names
Because your app executes inside a Docker container, it is really important to use Path.absname/2 to reference your .env files. The reason we mention this ad nauseam in these docs is because it's so critical.  Different release strategies may set the RELEASE_ROOT variable with or without a trailing slash, and that can mean that Dotenvy can't find your .env files.  It can be helpful to specify the :require_files option to Dotenvy.source/2 just to help alert you when the paths may be pointing to the wrong place, e.g.
env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

source!(
  [
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    System.get_env()
  ],
  require_files: [Path.absname(".env", env_dir_prefix)]
)
Environment Variables
Fly.io sets a handful of environment variables when it deploys an app (which you can inspect by running System.get_env() from an iex shell).  The following are the most significant environment variables used in the config files generated by the Dotenvy Generators:
	RELEASE_ROOT
	DATABASE_URL
	PHX_HOST
	PORT
	PHX_SERVER
	DNS_CLUSTER_QUERY
	SECRET_KEY_BASE

See the Phoenix page for getting a Phoenix app to run using Dotenvy.


  

    1Password

1Password is a popular password manager. The 1Password CLI utility is a convenient way to read sensitive data out of a password vault.  The approach taken here is similar to what other password tools may require.
The syntax for accessing values out of a 1Password item is this:
op://<vault-name>/<item-name>/[section-name/]<field-name>

Assuming you have the op command installed, you can execute it and other system commands in your .env files by using the $() syntax (available since version 1.0.0 of Dotenvy).  For example:
DB_PASSWORD=$(op read op://MyVault/FooDatabase/password);
API_KEY=$(op read op://MyVault/ImportantAPI/key);

See Dotenvy.Parser.parse/3 for examples on how to customize or restrict the functionality available inside the $() blocks.
Executing Shell commands prior to version 1.0.0
If you need your variables to be populated with the output of shell commands in versions of Dotenvy prior to version 1.0.0, you must do a bit more footwork and rely on System.shell/2.  Use this with caution!
If you're unable to upgrade to version 1.0.0 or the shell commands you are attempting to run are somehow not supported by the Dotenvy.Parser, please file a bug! And then try the following.
Create a shell script that exports the necessary ENV vars, e.g. secrets.sh:
#!/bin/bash
export DB_PASSWORD=$(op read op://MyVault/FooDatabase/password);
export API_KEY=$(op read op://MyVault/ImportantAPI/key);

Next, we need to execute this file AND return all the values!  We do this by using && in our shell command, e.g.
System.shell(~s'bash -c "source secrets.sh && env"')
When it's all put together in your config/runtime.exs it should look something like this:
# config/runtime.exs
import Config
import Dotenvy

env_dir_prefix = System.get_env("RELEASE_ROOT") || Path.expand("./envs")

{raw_envs, _} = System.shell(~s'bash -c "source #{Path.absname("secrets.sh", env_dir_prefix)} && env"')
{:ok, system_env_vars} = Dotenvy.Parser.parse(raw_envs)

source!([
    Path.absname(".env", env_dir_prefix),
    Path.absname(".#{config_env()}.env", env_dir_prefix),
    Path.absname(".#{config_env()}.overrides.env", env_dir_prefix),
    system_env_vars
  ])
See Also
You may need to make some adjustments to your security settings your your command line utility doesn't prompt you for authorization each time you run it:
https://apple.stackexchange.com/questions/442220/how-to-stop-iterm2-requiring-being-granted-access-every-time


  

    Philosophy

Dotenvy was born out of a desire to make Elixir's configuration more explicit and cleave more closely to how many other languages and frameworks are configured. Many Elixir apps contain a mishmash of compile-time and runtime configuration that can be confusing or unpredictable. Sometimes the choice of where to configure a particular thing is more the product of convenience rather than the result of thoughtful deliberation. Dotenvy attempts to take the guesswork out of the configuration by encouraging us to be intentional and explicit with our app and the configuration values it needs.
Taking its cue from the 12-factor app, Dotenvy's modus operandi is as follows:
	If something can be configured at runtime, then it should be configured at runtime. Keep the compile-time configuration for things that must be configured at compile-time.

	Store configuration in the environment. Elixir apps configured with Dotenvy tend to be smaller and lighter because they are not as burdened with conditionals. Let the app be the app and let the environment provide the configuration it needs.

	Have a clean contract with the underlying operating system: the config/runtime.exs can leverage Dotenvy.env!/2 to specify exactly the values it needs in a simple declarative way.

	Be suitable for deployment on modern cloud platforms: any system/software that can provide environment variables will work with Dotenvy.

	Minimize divergence between development and production: you can end up with a version of the app that is identical across all environments.




  

    Dotenv File Format

.env files (a.k.a. "dotenv") store key-value pairs in a format descended from
simple bash files that exported environment variables.
This implementation cleaves closely to the format described by the original dotenv package, but it is not a direct match (by design).
Typically, a dotenv (.env) file is formatted into simple key-value pairs:
S3_BUCKET=YOURS3BUCKET
SECRET_KEY=YOURSECRETKEYGOESHERE
Optionally, you may add export in front of each line so you can source the file in Bash:
export S3_BUCKET=YOURS3BUCKET
export SECRET_KEY=YOURSECRETKEYGOESHERE
Variable Names
For the sake of portability (and sanity), environment variable names must consist solely of letters, digits, and the underscore ( _ ) and must not begin with a digit. In regex-speak, the names must match the following pattern:
[a-zA-Z_]+[a-zA-Z0-9_]*
Example variable names
DATABASE_URL  # ok  
foobar        # ok  
NO-WORK       # <-- invalid !!!
ÜBER          # <-- invalid !!!
2MUCH         # <-- invalid !!!
Values
Values are to the right of the equals sign. They may be quoted. Using single quotes will prevent variables from being interpolated and any command substitution will not be executed.
SIMPLE=xyz123
INTERPOLATED="Multiple\nLines and variable substitution: ${SIMPLE}"
NON_INTERPOLATED='raw text without variable interpolation'
MULTILINE = """
long text here,
e.g. a private SSH key
"""
Escape Sequences
The following character strings will be interpreted (i.e. escaped) as specific codepoints in the same way you would expect if the values were assigned inside a script. Remember: when a text file is read, it is read as a series of utf8 encoded code points. Character sequences like \n have no special meaning until they are "escaped" and a combination of codepoints is replaced by a single codepoint.
	\n Linefeed (aka newline); <<92, 110>> -> <<10>>
	\r Carriage return; <<92, 114>> -> <<13>>
	\t Tab; -> <<92, 116>> -> <<9>>
	\f Form feed; -> <<92, 102>> -> <<12>>
	\b Backspace; -> <<92, 98>> -> <<8>>
	\" Double-quote; ->  <<92, 34>> -> <<34>>
	\' Single-quote; -> <<92, 39>> -> <<39>>
	\\ Backslash; -> <<92, 92>> -> <<92>>
	\uFFFF Unicode escape (4 hex characters to denote the codepoint)

If a backslash precedes any other character, the backslash will be ignored and removed from output.
Interpolation (a.k.a. Variable Substitution)
Values left unquoted or wrapped in double-quotes will interpolate variables in the ${VAR} syntax. This can be useful for referencing existing system environment variables or to reference variables previously parsed.
For example:
USER=admin
EMAIL=${USER}@example.org
DATABASE_URL="postgres://${USER}@localhost/my_database"
CACHE_DIR=${PWD}/cache
Multi-line values (e.g. private keys) can use the triple-quoted heredoc syntax:
PRIVATE_KEY="""
-----BEGIN RSA PRIVATE KEY-----
...
HkVN9...
...
-----END DSA PRIVATE KEY-----
"""
Non-Interpolated
If your values must retain ${} in their output, wrap the value in single quotes, e.g.:
PASSWORD='!@G0${k}k'
MESSAGE_TEMPLATE='''
    Hello ${PERSON},

    Nice to meet you!
'''
Comments
The hash-tag # symbol denotes a comment when on its own line or when it follows a quoted value.  It is not treated as a comment when it appears within quotes.
# This is a comment
SECRET_KEY=YOURSECRETKEYGOESHERE # also a comment
SECRET_HASH="something-with-a-hash-#-this-is-not-a-comment"
Command Substitution
You need to add the output of a shell command in one of your variables? Simply add it with $(your_command) (available since version 1.0.0). This is helpful when tying into 3rd party password managers, e.g. 1Password.
API_KEY=$(op read op://MyVault/SomeService/api_key)
As with interpolated variables, the $() syntax only triggers the execution of a shell command when it appears within double-quotes or in a line without quotes. If you quote your value with single quotes, the values will remain as-is.
If you need more control over how (or if) commands within the $() syntax are executed, the Dotenvy.Parser.parse/3 function supports 2 options: :sys_cmd_fn and :sys_cmd_opts. You can leverage these to do things such as perform linting of the .env files or disable system commands altogether.


  

    Configuration Providers

Configuration providers are most often invoked in the context of releases. Although they can solve certain problems that arise in production deployments, they tend to be an awkward fit for regular day-to-day development. Dotenvy seeks to normalize how configuration is loaded across environments, so having different strategies and functionality depending on how you run your app is antithetical. We do not want some code that runs only in certain environments and not in others: it can make for untested or untestable code!
Secondly, configuration providers sometimes shift the task of "shaping" the configuration out of Elixir and into some static representation (e.g. JSON or TOML). The allure of a straight-forward static file is deceiving because there is no easy way to delineate Elixir-specific subtleties. such as distinguishing between keyword lists and maps. For example, how do you distinguish between a keyword list and a map in TOML?  Or how to differentiate a tuple and a list in JSON? And how can you indicate a map with string keys or atom keys?
The takeaway is that When configuration providers "solve" one problem, they often create another and mental friction accumulates.  It makes sense to handle problems about Elixir-specific data types inside Elixir instead of trying to cram a solution into some unsuspecting static file format.
For these reasons, Dotenvy does not rely on configuration providers; .env files are an easier lingua franca.  As of Dotenvy version 1.0.0 and its support of shell commands, it's easier than ever to populate environment variables by using standard CLI tools to read values from password managers or other services.  See the page on 1Password for and example.


  

    Generators

The Dotenvy Generators package includes mix tasks which are designed to spin up new Elixir applications that leverage Dotenvy to read environment variables. This is one of the easiest ways to see Dotenvy in action.
The most important tasks include:
	dot.new: a variant of the humble mix new task
	phx.new: an alternate of the Phoenix mix phx.new task, used to spin up Phoenix applications

The generators are still being developed, so they may in the process of being updated/improved.
Installation
In order to install the Dotenvy generator scripts, you need to run two commands from your terminal: one to remove the phx_new generators (if present), and one to install the dotenvy_generators.  Run the following two commands:
mix archive.uninstall phx_new
mix archive.install hex dotenvy_generators
Once this has executed successfully, you should see the dot.new task as one of the available tasks when you run mix help.
We have to uninstall the phx_new generators because dotenvy_generators uses the same names for its mix tasks (this may change later... stay tuned).
See also the dedicated instructions in the Dotenvy Generators package.


  

    
Dotenvy behaviour
    



      
Dotenvy is an Elixir port of the original dotenv Ruby gem.
It is designed to help applications follow the principles of
the 12-factor app and its recommendation to store
configuration in the environment.
Unlike other configuration helpers, Dotenvy enforces no convention for the naming
of your files: .env is a common choice, you may name your configuration files whatever
you wish.
See the Getting Started page for more info.

      


      
        Summary


  
    Types
  


    
      
        input_source()

      


        An input source may be either a path to an env file or a map with string keys
and values, e.g. "envs/.env" or %{"FOO" => "bar"}. This allows users to
specify a list of env files interspersed with other values from other sources,
most commonly System.get_env().



    





  
    Callbacks
  


    
      
        parse(contents, vars, opts)

      


        A parser implementation should receive the contents read from a file,
a map of vars (with string keys, as would come from System.get_env/0),
and a keyword list of opts.



    





  
    Functions
  


    
      
        env!(variable, type \\ :string)

      


        Reads the given env variable and converts its value to the given type.



    


    
      
        env!(variable, type, default)

      


        Reads an env variable and converts its output or returns a default value.



    


    
      
        source(files, opts \\ [])

      


        Like its Bash namesake command, source/2 accumulates values from the given input(s).
The accumulated values are stored via a side effect function to make them available
to the env!/2 and env!/3 functions.



    


    
      
        source!(files, opts \\ [])

      


        As source/2, but returns a map on success or raises on error.



    





      


      
        Types


        


  
    
      
    
    
      input_source()



        
          
        

    

  


  

      

          @type input_source() :: String.t() | %{optional(String.t()) => String.t()}


      


An input source may be either a path to an env file or a map with string keys
and values, e.g. "envs/.env" or %{"FOO" => "bar"}. This allows users to
specify a list of env files interspersed with other values from other sources,
most commonly System.get_env().

  


        

      

      
        Callbacks


        


  
    
      
    
    
      parse(contents, vars, opts)



        
          
        

    

  


  

      

          @callback parse(contents :: binary(), vars :: map(), opts :: keyword()) ::
  {:ok, map()} | {:error, any()}


      


A parser implementation should receive the contents read from a file,
a map of vars (with string keys, as would come from System.get_env/0),
and a keyword list of opts.
See Dotenvy.Parser for the default implementation of this callback.

  


        

      

      
        Functions


        


    

  
    
      
    
    
      env!(variable, type \\ :string)



        
          
        

    

  


  

      

          @spec env!(variable :: binary(), type :: Dotenvy.Transformer.conversion_type()) ::
  any() | no_return()


      


Reads the given env variable and converts its value to the given type.
This function attempts to read a value from a local data store of sourced values.
This function may raise an error because type conversion is delegated to
Dotenvy.Transformer.to!/2 -- see its documentation for a list of supported types.
Examples
iex> env!("PORT", :integer)
5432
iex> env!("ENABLED", :boolean)
true

  



  
    
      
    
    
      env!(variable, type, default)


        (since 0.3.0)


        
          
        

    

  


  

      

          @spec env!(
  variable :: binary(),
  type :: Dotenvy.Transformer.conversion_type(),
  default :: any()
) :: any() | no_return()


      


Reads an env variable and converts its output or returns a default value.
Use env!/2 when possible
env!/2 is recommended over env!/3 because it creates a stronger contract with
the environment: your app literally will not start when required env variables are missing.
If the given variable is set, its value is converted to the given type.
The provided default value is only used when the variable is not set
(i.e. when it does not exist).
The default value is returned as-is, without conversion. This allows
greater control of the output.
Conversion is delegated to Dotenvy.Transformer.to!/2, which may raise an error.
See its documentation for a list of supported types.
This function attempts to read a value from a local data store of sourced values.
Examples
iex> env!("PORT", :integer, 5432)
5433

iex> env!("NOT_SET", :boolean, %{not: "converted"})
%{not: "converted"}

iex> System.put_env("HOST", "")
iex> env!("HOST", :string!, "localhost")
** (RuntimeError) Error converting HOST to string!: non-empty value required

  



    

  
    
      
    
    
      source(files, opts \\ [])



        
          
        

    

  


  

      

          @spec source(inputs :: input_source() | [input_source()], opts :: keyword()) ::
  {:ok, %{optional(String.t()) => String.t()}} | {:error, any()}


      


Like its Bash namesake command, source/2 accumulates values from the given input(s).
The accumulated values are stored via a side effect function to make them available
to the env!/2 and env!/3 functions.
Think of source/2 as a merging operation which can accept maps (like Map.merge/2)
or paths to env files.
Inputs are processed from left to right so that values can be overridden by each
subsequent input. As with Map.merge/2, the right-most input takes precedence.
Options
	:parser module that implements the Dotenvy.parse/3 callback. Default: Dotenvy.Parser

	:require_files specifies which of the given files (if any) must be present.
When true, all the listed files must exist.
When false, none of the listed files must exist.
When some of the files are required and some are optional, provide a list
specifying which files are required. If a file listed here is not included
in the function's files argument, it is ignored. Default: false

	:side_effect an arity 1 function called after successfully parsing inputs.
The default is an internal function that stores the values inside a process dictionary so
the values are available to the env!/2 and env!/3 functions. This option
is overridable to facilitate testing. Changing it is not recommended.


All other options are passed through to Dotenvy.Parser.parse/3, e.g. :sys_cmd_fn.
Examples
The simplest implementation is to parse a single file by including its path:
iex> Dotenvy.source(".env")
{:ok, %{
  "TIMEOUT" => "5000",
  "DATABASE_URL" => "postgres://postgres:postgres@localhost/myapp",
  # ...etc...
  }
}
More commonly, you will source multiple files (often based on the config_env())
and you will defer to pre-existing system variables. The most common pattern looks like this:
  iex> Dotenvy.source([
    "#{config_env()}.env",
    "#{config_env()}.override.env",
    System.get_env()
  ])
In the above example, the prod.env, dev.env, and test.env files would be version-controlled,
but the *.override.env variants would be ignored, giving developers the ability to override
values without needing to modify versioned files.
Give Precedence to System Envs!
Don't forget to include System.get_env() as the final input to source/2 so that
system environment variables take precedence over values sourced from .env files.
When you run a shell command like ❯ LOG_LEVEL=debug mix run, your expectation is probably that
the LOG_LEVEL variable would be set to debug, overriding whatever may have been defined
in your sourced .env files. Similarly, you may export env vars in your Bash profile.
System env vars are not granted precedence automatically: you must explicitly include
System.get_env() as the final input to source/2.
If your env files are making use of variable substitution based on system env vars,
e.g. ${PWD} (see the Dotenv File Format), then you
would need to specify System.get_env() as the first argument to source/2.
For example, if your .env references the system HOME variable:
  # .env
  CACHE_DIR=${HOME}/cache
then your source/2 command would need to make the system env vars available
to the parser by including them as one of the inputs, e.g.
  iex> Dotenvy.source([System.get_env(), ".env"])
In some cases, you may wish to reference the system vars both before and after
your own .env files, e.g.
  iex> Dotenvy.source([System.get_env(), ".env", System.get_env()])
or you may wish to cherry-pick which variables you need to make available for
variable substitution:
  iex> Dotenvy.source([
    %{"HOME" => System.get_env("HOME")},
    ".env",
    System.get_env()
  ])
This syntax favors explicitness so there is no confusion over what might have been
"automagically" accumulated.
If you need to make any variables parsed from your files available elsewhere in
the application via System.get_env/2, then you can call System.put_env/1 on
the output of Dotenvy.source/2, e.g.
iex> {:ok, parsed_vars} = Dotenvy.source([".env", System.get_env()])
iex> System.put_env(parsed_vars)

  



    

  
    
      
    
    
      source!(files, opts \\ [])



        
          
        

    

  


  

      

          @spec source!(files :: binary() | [binary()], opts :: keyword()) ::
  %{optional(String.t()) => String.t()} | no_return()


      


As source/2, but returns a map on success or raises on error.

  


        

      


  

    
Dotenvy.Parser 
    



      
This module handles the parsing of the contents of .env files into maps with
string keys. See Dotenv File Format for details
on the supported file format.
This implementation uses parsing over regular expressions for most of its work.

      


      
        Summary


  
    Functions
  


    
      
        parse(contents, vars \\ %{}, opts \\ [])

      


        Parse the given contents, substituting and merging with the given vars.



    





      


      
        Functions


        


    

    

  
    
      
    
    
      parse(contents, vars \\ %{}, opts \\ [])



        
          
        

    

  


  

Parse the given contents, substituting and merging with the given vars.
Examples
If you wish to disable or limit support for executing system commands (i.e. those inside $()),
you can provide a custom :sys_cmd_fn option. For example, to disable the feature altogether:
iex> Dotenvy.Parser.parse(contents, %{}, sys_cmd_fn: fn _cmd, _args, _opts -> {"", 0} end)
If you wish to limit the available commands, you can customize your function, e.g.
iex> Dotenvy.Parser.parse(contents, %{}, sys_cmd_fn: fn
  "op", args, opts -> System.cmd("op", args, opts)
  _cmd, _args, _opts -> raise "Command not allowed"
end)
Options
	:sys_cmd_fn an arity 3 function returning a tuple matching the spec for
the System.cmd/3 function: the
first element is the raw output and the second represents the exit status (0
on success). Default: System.cmd/3
	:sys_cmd_opts keyword list of options passed as the 3rd arg to the :sys_cmd_fn.


  


        

      


  

    
Dotenvy.Transformer 
    



      
This module provides functionality for converting string values to specific Elixir data types.
These conversions were designed to operate on system environment variables, which
always store string binaries.

      


      
        Summary


  
    Types
  


    
      
        conversion_type()

      


        The conversion type specifies the target data type to which a string will be converted.
For example, :integer would indicate a transformation of "12" to 12.



    





  
    Functions
  


    
      
        to!(str, callback)

      


        Converts strings into Elixir data types with support for nil-able values. Raises on error.



    





      


      
        Types


        


  
    
      
    
    
      conversion_type()



        
          
        

    

  


  

      

          @type conversion_type() ::
  :atom
  | :atom?
  | :atom!
  | :boolean
  | :boolean?
  | :boolean!
  | :charlist
  | :charlist?
  | :charlist!
  | :integer
  | :integer?
  | :integer!
  | :float
  | :float?
  | :float!
  | :existing_atom
  | :existing_atom?
  | :existing_atom!
  | :module
  | :module?
  | :module!
  | :string
  | :string?
  | :string!
  | (String.t() -> any())


      


The conversion type specifies the target data type to which a string will be converted.
For example, :integer would indicate a transformation of "12" to 12.
The following types are supported:
	:atom - converts to an atom. An empty string will be the atom :"" (!).

	:atom? - converts to an atom. An empty string will be considered nil

	:atom! - converts to an atom. An empty string will raise.

	:boolean - "false", "0", or an empty string "" will be considered boolean false. Any other non-empty value is considered true.

	:boolean? - as above, except an empty string will be considered nil

	:boolean! - as above, except an empty string will raise.

	:charlist - converts string to charlist.

	:charlist? - converts string to charlist. Empty string will be considered nil.

	:charlist! - as above, but an empty string will raise.

	:integer - converts a string to an integer. An empty string will be considered 0.

	:integer? - as above, but an empty string will be considered nil.

	:integer! - as above, but an empty string will raise.

	:float - converts a string to an float. An empty string will be considered 0.

	:float? - as above, but an empty string will be considered nil.

	:float! - as above, but an empty string will raise.

	:existing_atom - converts into an existing atom. Raises error if the atom does not exist.

	:existing_atom? - as above, but an empty string will be considered nil.

	:existing_atom! - as above, but an empty string will raise.

	:module - converts a string into an Elixir module name. Raises on error.

	:module? - as above, but an empty string will be considered nil.

	:module! - as above, but an empty string will raise.

	:string - no conversion (default)

	:string? - empty strings will be considered nil.

	:string! - as above, but an empty string will raise.

	custom function - see below.


Custom Callback function
When you require more control over the transformation of your value than is possible
with the types provided, you can provide an arity 1 function in place of the type.

  


        

      

      
        Functions


        


  
    
      
    
    
      to!(str, callback)



        
          
        

    

  


  

      

          @spec to!(str :: binary(), type :: conversion_type()) :: any()


      


Converts strings into Elixir data types with support for nil-able values. Raises on error.
Each type determines how to interpret the incoming string, e.g. when the type
is :integer, an empty string is considered a 0; when :integer? is the type,
and empty string is converted to nil.
Remember:
	The type should use a ? suffix when an empty string should be considered nil (a.k.a. a "nullable" value).
	The type should a ! suffix when an empty string is not allowed. Use this when values are required.

Types
See the Dotenvy.Transformer.conversion_type/0 for a description of valid
conversion types.
Examples
iex> to!("debug", :atom)
:debug
iex> to!("", :boolean)
false
iex> to!("", :boolean?)
nil
iex> to!("5432", :integer)
5432
iex> to!("DateTime", :module)
DateTime
iex> to!("foo", fn val -> val <> "bar" end)
"foobar"
iex> Dotenvy.Transformer.to!("Oops", :float)
  ** (Dotenvy.Error) Unparsable as float
  (dotenvy 1.0.0) lib/dotenvy/transformer.ex:165: Dotenvy.Transformer.to!/2

  


        

      


  

    
Dotenvy.Error exception
    



      
This error module can be useful when writing your own custom conversion
functions because special contextual information will be included with any
errors.
Examples
Let's say your configuration needs to supply one of a set of possible values
(i.e. an enum). We can define a custom function to support this and pass it
as the second argument to Dotenvy.env!/2
# runtime.exs
import Config
import Dotenvy

size_enum = fn
  "large" -> :large
  "small" -> :small
  _ ->
    raise Dotenvy.Error, message: "allowed size_enum values are large or small"
end

config :myapp, :some_bool, env!("SIZE", size_enum)

